
Torsionsschwingungen  
mit Übertragungsmatrizen berechnen 

Grundlagen 

Im Abschnitt  5.3.5 wurde das Modell eines rotierenden Zweimassensystems entwickelt und 
aus der Lösung des Differentialgleichungssystems u. a. eine Formel für die Eigenkreisfrequenz 
des Zweimassensystems abgeleitet. Oft interessiert aber das Schwingungsverhalten eines rotie-
renden Systems lediglich im stationären, eingeschwungenen Zustand. Auch haben die Systeme 
nicht selten mehr als zwei Drehmassen. In einem solchen Fall ist das im Folgenden dargestellte 
Verfahren für die Berechnung der Eigenkreisschwingungen gut geeignet. Es wird am Beispiel 
des im Bild 5.11 gezeigten Rotationssystems, bestehend aus drei Drehmassen mit den Träg-
heitsmomenten Ji, die durch zwei als masselos angenommene Wellen mit den Federkonstanten 
ci verbunden sind, beschrieben. 

 

Bild 0.1  Rotierendes System mit drei Massen 

Das Verfahren beruht auf der Aufteilung des Gesamtsystems in Segmente (oft auch Felder 
genannt). Die Schnittstellen zwischen den Segmenten werden durch die dort wirkenden Zu-
standsgrößen, im vorlegenden Beispiel Drehwinkel φ und Moment M, beschrieben. Den Zu-
sammenhang zwischen den Zustandsgrößen am rechten und am linken Rand eines Segments 
liefert eine Matrix (Übertragungsmatrix), die für jedes Segment aus Gleichgewichtsbeziehun-
gen und kinematischen Zwangsbedingungen bestimmt werden kann. Im Bild 5.12 ist das seg-
mentierte System mit den Zustandsgrößen dargestellt. Die Indizes der Trägheitsmomente und 
der Federkonstanten sind darin der Bezeichnung der Segmente k angepasst. 

 

Bild 0.2  Schnittstellen und Segmente des Rotationssystems  

Einführend werden die Grundlagen des Verfahrens beschrieben [Göld67, HoDre94]. Für die 
Drehmasse (Scheibe) des Segments k im Bild 5.12 gilt 

1k kϕ ϕ −=  (5.51) 

Der Drehwinkel φ wird in der Wirkungsrichtung des Momentes M auf der linken Seite einer 
Scheibe als positiv angenommen. Damit ist 

1 0k k kM M J ϕ− − − ⋅ =  (5.52) 

Für die ungedämpfte Eigenschwingung ist bekanntlich 
2 2;j t j tA e A eω ωϕ ϕ ω ω ϕ⋅ ⋅= ⋅ = − ⋅ ⋅ = − ⋅  (5.53) 

Gl. (5.53) und (5.51) in (5.52) eingesetzt liefert 
2

1 1k k k kM M J ω ϕ− −= + ⋅ ⋅  (5.54) 

Durch (5.51) und (5.54) werden damit die Zustandsgrößen rechts von der Drehmasse im Seg-
ment k durch die Zustandsgrößen links von ihr ausgedrückt. In Matrizenschreibweise ergibt 
sich 

Die Matrix TM,k wird als Übertragungsmatrix der Drehmasse im Segment k bezeichnet. Die 
Vektoren zk und zk-1 sind die Zustandsvektoren der rechten und der linken Seite der Drehmasse 
oder allgemeiner des Segments k (Bild 5.12).  
Die Beziehung zwischen den Zustandsgrößen auf der linken und der rechten Seite einer Torsi-
onsfeder (Welle) mit der Federkonstanten ci lässt sich ebenfalls durch eine Matrizengleichung 
beschreiben. Für jede Welle des Segments k gilt das Momentengleichgewicht 

1k kM M −=  (5.57) 

und 

( )1 1k k k kM c ϕ ϕ− −= −  (5.58) 

Aus (5.57) und (5.59) folgt daher für die Torsionsfeder 

bzw. in Kurzschreibweise  

Ein Schwingungssystem gemäß Bild 5.12 kann man nun stufenweise wie folgt berechnen: 

0 1 1 0 2 2 1 3 3 2 4 4 3 5 5 4: : : : :→ = ⋅ → = ⋅ → = ⋅ → = ⋅ → = ⋅z z T z z T z z T z z T z z T z  

Die Randbedingungen z0 bzw. z5 sind durch die konkreten Bedingungen vorgegeben. Für das 
im Bild 5.12 gezeigte Beispiel sind die Elemente des Zustandsvektors z0 folgendermaßen defi-
niert: M0 = 0; φ0 unbekannt.  

Berechnungsbeispiel 

Die Umsetzung des beschriebenen Verfahrens in Maple wird anhand des in Bild 5.12 darge-
stellten Rotationssystems gezeigt. 
> interface(displayprecision=3): 

Notierung der Übertragungsmatrizen Tk: 

> T[1]:= Matrix([[1,0],[J[1]*omega^2, 1]]); 

 
> T[2]:= Matrix([[1, -1/c[1]],[0, 1]]); 

 
> T[3]:= Matrix([[1,0],[J[2]*omega^2, 1]]): 
> T[4]:= Matrix([[1, -1/c[2]],[0, 1]]): 
> T[5]:= Matrix([[1,0],[J[3]*omega^2, 1]]): 

Vorgabe der Bedingungen am linken Rand: 
> M[0]:=0: 
> z[0]:= Vector([phi[0], M[0]]):   

Berechnung der Zustandsvektoren 

In der for-Schleife werden die Vektoren aller Zustandsgrößen des Schwingungssystems ermit-
telt.  
> for k from 1 to 5 do 
    z[k]:= T[k].z[k-1]; 
    end do;  

Das Moment am rechten Rand des Schwingungssystems wird durch die 2. Komponente des 
zuletzt berechneten Zustandsvektors z5 = z[5] beschrieben. Der folgende Befehl stellt dieses 
nach Potenzen von ω geordnet dar. 
> collect(z[5][2], omega); 

Berechnung der Eigenkreisfrequenzen 

Weil die rechte Seite der Drehmasse 3 frei ist, muss das Moment M an dieser Stelle Null sein. 
Unter Verwendung der Bedingung z[5][2] = 0 können nun die Eigenkreisfrequenzen berechnet 
werden. Um zu einigermaßen übersichtlichen Ausdrücken zu gelangen, werden zuerst die 
Quadrate der Eigenkreisfrequenzen bestimmt. Dazu wird ω2 in zr[5][2] durch die Variable om2 
ersetzt und dann die neue Gleichung w nach dieser Variablen aufgelöst. 
> w:= subs(omega^2=om2, z[5][2]): 

Quadrate der Eigenkreisfrequenzen: 
> EWq:= solve(w, om2); 

 
Die gleiche Lösung erhält man, wenn im Befehl solve als erstes Argument der unveränderte 
Ausdruck z[5][2] und als zweites omega^2 angegeben wird, allerdings mit der Maple-War-
nung, dass diese Notierungsform nicht zu empfehlen ist: 
Warning, solving for expressions other than names or functions is not 
recommended. 
Die Eigenkreisfrequenzen erhalten wir nun durch Berechnung der Quadratwurzeln aller Ele-
mente der Lösungsfolge EWq. Zu diesem Zweck wird EWq als Liste notiert und der Befehl 
map benutzt. Dieser bewirkt, dass die Funktion sqrt auf alle Elemente der Liste [EWq] ange-
wendet wird. Das Ergebnis ist dann die Liste EW.  
> EW:= map(sqrt, [EWq]); 

Das Ergebnis unterscheidet sich in der Darstellung von der für EWq nur dadurch, dass an die 
Ausdrücke für die Quadrate der Eigenkreisfrequenzen der Exponent ½ angehängt ist.  
Numerische Auswertung 

Für das im Bild 5.11 dargestellte Schwingungssystem seien folgende Parameterwerte bekannt: 
> param3:= J[1]=1, J[2]=2, J[3]=1, c[1]=100, c[2]=200:  

Mit dem Parametersatz param3 liefert die Auswertung von EW folgende Werte für die Eigen-
kreisfrequenzen: 
> EW3:= evalf(subs(param3, EW)); 

 
Das im Bild 5.11 dargestellte System mit drei Drehmassen hat demnach drei Eigenfrequenzen, 
wenn ω1 = 0 mitgezählt wird [HoDre94, S. 191]. Im Folgenden interessieren jedoch nur die 
beiden Werte, die größer als Null sind. Der kleinere davon ist die Grundschwingung mit der 
Kreisfrequenz ω = 11,042 s-1, der größere die Oberschwingung mit ω = 18,113 s-1. Zur Ver-
einfachung zukünftiger Operationen mit Grund- und Oberschwingung werden die betreffenden 
Werte separaten Variablen zugewiesen. 
> GS:= EW3[3]:  OS:= EW3[2]: 

Ermittlung der Amplituden Ai und der Amplitudenverhältnisse der Drehmassen  

Zur Ermittlung der Amplituden wird auf die 1. Komponenten der oben berechneten Zustands-
vektoren zugegriffen. 
Amplitude von Drehmasse J1: 
> A1:= z[1][1]; 

 
Amplitude von Drehmasse J3: 
> A2:= z[3][1]; 

 
Amplitude von Drehmasse J5 
> A3:= z[5][1]; 

 
Daraus ergeben sich die symbolischen Ausdrücke für die Amplitudenverhältnisse vi . Alle 
Amplituden werden auf A1 = φ0 bezogen. 
> v[1]:= 1: 
> v[2]:= simplify(A2/A1); 

 
> v[3]:= simplify(A3/A1); 

> v[3]:= collect(v[3], omega); 

 

Darstellung der Eigenschwingformen 

Eigenschwingformen (Eigenformen) beschreiben die Form einer Schwingung bei einer be-
stimmten Eigenfrequenz. Sie ergeben sich durch Einsetzen der betreffenden Frequenz in die 
Ausdrücke für die Amplitudenverhältnisse. Die Eigenschwingformen sind bei der Bewertung 
des Einflusses von Parameteränderungen auf die Eigenfrequenzen und für die Anlalyse der 
Anregbarkeit erzwungener Schwingungen hilfreich [HoDre94]. 
Eigenschwingform der Grundschwingung: 
> V1:= evalf(subs([param3,omega=GS], [v[1],v[2],v[3]])); 

 
Eigenschwingform der Oberschwingung: 
> V2:= evalf(subs([param3,omega=OS], [v[1],v[2],v[3]])); 

 
Positionen der Drehmassen: 
> x:= [0, 1, 3]: 

Graphische Darstellung Eigenschwingformen der Grundschwingung (schwarz) und der Ober-
schwingung (blau): 
> with(plots): multiple(plot,[x,V1,color=black],[x,V2, color=blue], 
       title="Eigenschwingformen \n von Grund- und Oberschwingung", 
       titlefont=[HELVETICA,12]); 

 

Die Nulldurchgänge der Kurven geben die Stellen der verbindenden Welle an, an denen keine 
Verdrehung durch die betreffende Schwingung auftritt (Schwingungsknoten). Für die Grund-
schwingung ist die Differenz der Amplitudenverhältnisse zwischen Drehmasse 1 und 2 wesent-
lich größer als zwischen Drehmasse 2 und 3 (Bild 5.11). Die Federsteifigkeit c1 hat demnach 
auf die Frequenz der Grundschwingung einen stärkeren Einfluss als c2. 

Ermittlung der Momente 

Die Momente in den Wellenabschnitten sind die 2. Größen in den Zustandsvektoren zk .  

Moment zwischen den Drehmassen 1 und 2: 
> M1:= z[1][2]; 

 
Moment zwischen den Drehmassen 2 und 3: 
> M2:= collect(z[3][2], omega); 

 
Momentenamplituden für die Grundschwingung: 
> evalf(subs([param3, omega=GS], [M1,M2])); 

 
Momentenamplituden für die Oberschwingung: 
> evalf(subs([param3, omega=OS], [M1,M2])); 

 
Ist Drehwinkel φ0 bekannt, dann kann man die exakte Größe der Drehmomente berechnen. 

